

# Conjunction Capers

A TLA+ Truffle



Image: Amazon

Ron Pressler, October 2020

**“Just as natural pearls grow from grains of sand that have irritated oysters, these programming pearls have grown from real problems that have irritated programmers. The programs are fun, and they teach important programming techniques and fundamental design principles.”**

**Jon Bentley, Programming Pearls, Communications of the ACM**



# TLA<sup>+</sup> Truffle

## In the spirit of Programming Pearls and Graphics Gems

- an instructive example of program calculation or proof, or
- a nifty presentation of an old or new data structure, or
- an interesting application or programming technique.

**“polished, elegant, instructive, entertaining”**

Richard Bird, *How to Write a Functional Pearl*

# The Technique



Image: Amazon




# The Canonical TLA Formula

## A State-Machine Specification

$$Init \wedge \square [Next]_{vars} \wedge Fairness$$


# The Meaning of a TLA Formula





# The Meaning of a TLA Formula

Conjunction is Composition



# The Meaning of a TLA Formula

## Conjunction

$$\frac{F \Rightarrow A \quad G \Rightarrow B}{F \wedge G \Rightarrow A}$$

$$\frac{F \Rightarrow A \quad G \Rightarrow B}{F \wedge G \Rightarrow B}$$



# Conjunction of Canonical TLA Formulas

$Init \wedge \square [Next]_{vars} \wedge Fairness$

$\wedge$

$Init \wedge \square [Next]_{vars} \wedge Fairness$

# Conjunction of Canonical TLA Formulas

$\wedge \textit{Init}$

$\wedge \textit{Init}$

$\wedge \square [\textit{Next}]_{\textit{vars}}$

$\wedge \square [\textit{Next}]_{\textit{vars}}$

$\wedge \textit{Fairness}$

$\wedge \textit{Fairness}$



# Conjunction of Canonical TLA Formulas

$\wedge \textit{Init}$                        $\wedge \textit{Init}$

$\wedge \square [\textit{Next}]_{\textit{vars}}$     $\wedge \square [\textit{Next}]_{\textit{vars}}$

$\wedge \textit{Fairness}$                  $\wedge \textit{Fairness}$

$\equiv \textit{Init} \wedge \textit{Init} \wedge \square [???]_{???} \wedge \textit{Fairness} \wedge \textit{Fairness}$

# Conjunction of Canonical TLA Formulas

$$(1) \quad (p \vee q) \wedge (r \vee s) \equiv \vee p \wedge r \\ \vee p \wedge s \\ \vee q \wedge r \\ \vee q \wedge s$$

$$(2) \quad \square A \wedge \square B \equiv \square (A \wedge B)$$

$$(3) \quad \square [A]_{vars} \equiv \square (A \vee \text{UNCHANGED } vars) \\ \equiv \square (A \vee vars' = vars)$$

# Conjunction of Canonical TLA Formulas

$$\square [Next]_{vars} \wedge \square [Next]_{vars}$$

$$\equiv \square ( \vee Next \wedge Next \\ \vee Next \wedge \text{UNCHANGED } vars \\ \vee Next \wedge \text{UNCHANGED } vars \\ \vee \text{UNCHANGED } vars \wedge \text{UNCHANGED } vars )$$

# Conjunction of Canonical TLA Formulas

$$\square [Next]_{vars} \wedge \square [Next]_{vars}$$

$$\equiv \square [ \begin{array}{l} \vee Next \wedge Next \\ \vee Next \wedge \text{UNCHANGED } vars \\ \vee Next \wedge \text{UNCHANGED } vars \end{array} ]_{\langle vars, vars \rangle}$$



# Conjunction of Canonical TLA<sup>+</sup> Formulas

$$\textit{Compose}(\textit{Next}A, \textit{vars}A, \textit{Next}B, \textit{vars}B) \triangleq$$

$$\vee \textit{Next}A \wedge \textit{Next}B$$

$$\vee \textit{Next}A \wedge \textit{UNCHANGED } \textit{vars}B$$

$$\vee \textit{Next}B \wedge \textit{UNCHANGED } \textit{vars}A$$

# Conjunction of Canonical TLA<sup>+</sup> Formulas

$$\text{Compose}(\text{Next}A, \text{vars}A, \text{Next}B, \text{vars}B) \triangleq$$

$$\vee \text{Next}A \wedge \text{Next}B$$

$$\vee \text{Next}A \wedge \text{UNCHANGED } \text{vars}B$$

$$\vee \text{Next}B \wedge \text{UNCHANGED } \text{vars}A$$

$$x' = A$$

$$x' \in A$$

# Conjunction of Canonical TLA<sup>+</sup> Formulas

$$\textit{Compose}(\textit{NextA}, \textit{varsA}, \textit{NextB}, \textit{varsB}) \triangleq$$

$$\vee \textit{NextA} \wedge \textit{NextB}$$

$$\vee \text{UNCHANGED } \textit{varsB} \wedge \textit{NextA}$$

$$\vee \text{UNCHANGED } \textit{varsA} \wedge \textit{NextB}$$

# Conjunction of Canonical TLA<sup>+</sup> Formulas

$$\begin{aligned} \textit{Spec} &\triangleq \wedge \textit{Init1} \wedge \square[\textit{Next1}]_{v_1} \wedge \textit{Fairness1} \\ &\wedge \textit{Init2} \wedge \square[\textit{Next2}]_{v_2} \wedge \textit{Fairness2} \end{aligned}$$

$$\begin{aligned} \textit{Spec} &\triangleq \wedge \textit{Init1} \wedge \textit{Init2} \\ &\wedge \square[\textit{Compose}(\textit{Next1}, v_1, \textit{Next2}, v_2)]_{\langle v_1, v_2 \rangle} \\ &\wedge \textit{Fairness1} \wedge \textit{Fairness2} \end{aligned}$$



# Conjunction of Canonical TLA<sup>+</sup> Formulas

$$Next12 \triangleq Compose(Next1, v1, Next2, v2)$$

$$Next123 \triangleq Compose(Next12, \langle v1, v2 \rangle, Next3, v3)$$

$$Next1234 \triangleq Compose(Next123, \langle v1, v2, v3 \rangle, Next4, v4)$$

$$Next12345 \triangleq Compose(Next1234, \langle v1, v2, v3, v4 \rangle, Next5, v5)$$

$$Next123456 \triangleq Compose(Next12345, \langle v1, v2, v3, v4, v5 \rangle, Next6, v6)$$

$$Next1234567 \triangleq Compose(Next123456, \langle v1, v2, v3, v4, v5, v6 \rangle, Next7, v7)$$

$$vars \triangleq \langle v1, v2, v3, v4, v5, v6, v7 \rangle$$

$$Init \triangleq Init1 \wedge Init2 \wedge Init3 \wedge Init4 \wedge Init5 \wedge Init6 \wedge Init7$$

$$Next \triangleq Next1234567$$

$$Spec \triangleq Init \wedge \square[Next]_{vars}$$



# Example 1





# Behavioral Programming

David Harel, Assaf Marron, Gera Weiss, CACM, 2012

DOI:10.1145/2209249.2209270

**A novel paradigm for programming reactive systems centered on naturally specified modular behavior.**

BY DAVID HAREL, ASSAF MARRON, AND GERA WEISS

## Behavioral Programming

SPELLING OUT THE requirements for a software system under development is not an easy task, and translating captured requirements into correct operational software can be even harder. Many technologies (languages, modeling tools, programming paradigms) and methodologies (agile, test-driven, model-driven) were designed, among other things, to help address these challenges. One widely accepted practice is to formalize requirements in the form of use cases and scenarios. Our work extends this approach into using scenarios

To illustrate the naturalness of constructing systems by composing behaviors, consider how children may be taught, step-by-step, to play strategy games (See Gordon et al.<sup>14</sup>). For example, in teaching the game of Tic-Tac-Toe, we first describe rules of the game, such as:

**EnforceTurns:** To play, one player marks a square in a 3 by 3 grid with X, then the other player marks a square with O, then X plays again, and so on;

**SquareTaken:** Once a square is marked, it cannot be marked again;

**DetectXWin/DetectOWin:** When a player places three of his or her marks in a horizontal, vertical, or diagonal line, the player wins;

Now we may already start playing. Later, the child may infer, or the teacher may suggest, some tactics:

**AddThirdO:** After placing two Os in a line, the O player should try to mark the third square (to win the game);

**PreventThirdX:** After the X player marks two squares in a line, the O player should try to mark the third square (to foil the attack); and

**DefaultOMoves:** When other tactics are not applicable, player O should prefer the center square, then the cor-

# Behavioral Programming

## Tic-Tac-Toe

1. Board: At each step, an  $X$  or an  $O$  is marked on the board

VARIABLE  $board$

$v1 \triangleq \langle board \rangle$

$N \triangleq 3$

$Empty \triangleq \text{"-"}$

$Player \triangleq \{\text{"X"}, \text{"O"}\}$

$Mark \triangleq Player$

$Square \triangleq \{Empty\} \cup Mark$

$BoardType \triangleq board \in [(1..N) \times (1..N) \rightarrow Square]$

$BoardFull \triangleq \forall i, j \in 1..N : board[i, j] \neq Empty$

$Init1 \triangleq board = [i, j \in 1..N \mapsto Empty]$

$Next1 \triangleq \exists i, j \in 1..N, mark \in Mark : \wedge board[i, j] = Empty$   
 $\wedge board' = [board \text{ EXCEPT } ![i, j] = mark]$

$Board \triangleq Init1 \wedge \square[Next1]_{v1}$

$TicTacToe1 \triangleq Board$

Properties we can state at this point:

THEOREM  $TicTacToe1 \Rightarrow \square BoardType$

$OnceSetAlwaysSet \triangleq$

$\forall i, j \in 1..N : \square(\exists mark \in Mark : board[i, j] = mark \Rightarrow \square(board[i, j] = mark))$

THEOREM  $TicTacToe1 \Rightarrow OnceSetAlwaysSet$



# Behavioral Programming

## Tic-Tac-Toe

2. *EnforceTurns*: *X* and *O* play in alternating turns

VARIABLE *current*,

*turn* Necessary for some properties we may wish to state

$v2 \triangleq \langle v1, turn, current \rangle$

$Other(player) \triangleq \text{IF } player = \text{"X"} \text{ THEN "O" ELSE "X"}$

$Opponent \triangleq Other(current)$

$TurnType \triangleq \wedge current \in Player$   
 $\wedge turn \in Nat$

$Init2 \triangleq \wedge turn = 0$   
 $\wedge current = \text{"X"}$  *X starts*

$Next2 \triangleq \wedge turn' = turn + 1$   
 $\wedge current' = Opponent$   
 $\wedge \exists i, j \in 1 .. N : \wedge board[i, j] = Empty$   
 $\wedge board'[i, j] = current$

$EnforceTurns \triangleq Init2 \wedge \square[Next2]_{v2}$

$TicTacToe2 \triangleq TicTacToe1 \wedge EnforceTurns$

Properties we can state at this point:

THEOREM  $EnforceTurns \Rightarrow TurnType$

$Alternating \triangleq \square[current' \neq current]_{v2}$

THEOREM  $EnforceTurns \Rightarrow Alternating$

# Behavioral Programming

## Tic-Tac-Toe

### Statistics

State space progress (click column header for graph)

Time	Diameter	States Found	Distinct States
00:00:25	10	21,630	5,478
00:00:05	5	580	412
00:00:02	0	1	1

3. *DetectWin*: Detect win or draw and end game

VARIABLE *win*  
 $v3 \triangleq \langle v2, win \rangle$

*Result*  $\triangleq Player \cup \{ \text{"Draw"} \}$   
*WinType*  $\triangleq win \in \{ Empty \} \cup Result$   
*GameEnd*  $\triangleq win \in Result$

*Line*  $\triangleq \{ [i \in 1..N \mapsto \langle i, y \rangle] : y \in 1..N \}$  horizontal  
 $\cup \{ [i \in 1..N \mapsto \langle x, i \rangle] : x \in 1..N \}$  vertical  
 $\cup \{ [i \in 1..N \mapsto \langle i, i \rangle] \} \cup \{ [i \in 1..N \mapsto \langle i, N - i + 1 \rangle] \}$  diagonal

$f \circ g \triangleq [x \in \text{DOMAIN } g \mapsto f[g[x]]]$   
*BoardLine*(*line*)  $\triangleq board \circ line$

*Won*(*player*)  $\triangleq \exists line \in Line : BoardLine(line) = [i \in 1..N \mapsto player]$   
*NoWin*  $\triangleq \neg \exists player \in Player : Won(player)'$   
*StopGame*  $\triangleq board' = board$  UNCHANGED *board* - fails *TLC*

*Init3*  $\triangleq win = Empty$   
*Next3*  $\triangleq \vee \wedge win = Empty$   
 $\wedge \vee \exists player \in Player : Won(player)' \wedge win' = player$   
 $\vee NoWin \wedge BoardFull' \wedge win' = \text{"Draw"}$   
 $\vee NoWin \wedge \neg BoardFull' \wedge \text{UNCHANGED } win$   
 $\vee \wedge win \in Player$   
 $\wedge \text{UNCHANGED } win$   
 $\wedge StopGame$

*DetectWin*  $\triangleq Init3 \wedge \square [Next3]_{v3}$

*TicTacToe3*  $\triangleq TicTacToe2 \wedge DetectWin$

Properties we can state at this point:  
 THEOREM *DetectWin*  $\Rightarrow WinType$

*GameEndsWhenPlayerWins*  $\triangleq \square (win \in Player \Rightarrow \square [board' = board]_{-v3})$   
*GameEndsWhenPlayerWins*  $\triangleq \square [(win \in Player \Rightarrow \text{UNCHANGED } board)]_{v3}$   
 THEOREM *TicTacToe3*  $\Rightarrow GameEndsWhenPlayerWins$

*AtLeast5TurnsToWin*  $\triangleq win \neq Empty \Rightarrow turn \geq 2 * N - 1$   
 THEOREM *TicTacToe3*  $\Rightarrow \square (AtLeast5TurnsToWin)$

*GameEndsWhenBoardFull*  $\triangleq BoardFull \Rightarrow GameEnd$   
 THEOREM *TicTacToe3*  $\Rightarrow \square (GameEndsWhenBoardFull)$

# Behavioral Programming

## Tic-Tac-Toe

4. *AddThirdToWin*: Add third mark to win

So far, we've specified the rules of the game. Now we start adding tactic rules. This one says that if a player has two marks in a line they should place the third to win.

But we run into a problem: the tactics may be contradictory, and prioritization is required. *b*-threads can be prioritized, and we could simulate that mechanism with maps of boolean functions, but that would be overly clever, especially in a simple specification such as this. Instead, we'll order the rules by their priority, and explicitly model priorities. This means that new rules would need to be inserted in the sequence of rules into their right position.

$$\text{Count}(\text{mark}, \text{line}) \triangleq \text{Cardinality}(\{i \in 1 \dots N : \text{BoardLine}(\text{line})[i] = \text{mark}\})$$

$$\begin{aligned} \text{CanWin}(\text{player}) \triangleq & \exists \text{line} \in \text{Line} : \wedge \text{Count}(\text{player}, \text{line}) = N - 1 \\ & \wedge \text{Count}(\text{Empty}, \text{line}) = 1 \end{aligned}$$

$$\begin{aligned} \text{MarkLast}(\text{line}) \triangleq & \exists i \in 1 \dots N : \wedge \text{BoardLine}(\text{line})[i] = \text{Empty} \\ & \wedge \text{board}'[\text{line}[i]] = \text{current} \end{aligned}$$

$$v4 \triangleq v3$$

$$\text{Init4} \triangleq \text{TRUE}$$

$$\text{Next4} \triangleq \text{CanWin}(\text{current}) \Rightarrow$$

$$\exists \text{line} \in \text{Line} : \text{Count}(\text{current}, \text{line}) = N - 1 \wedge \text{MarkLast}(\text{line})$$

$$\text{Priority1} \triangleq \text{CanWin}(\text{current})$$

$$\text{AddThirdToWin} \triangleq \text{Init4} \wedge \square[\text{Next4}]_{v4}$$

$$\text{TicTacToe4} \triangleq \text{TicTacToe3} \wedge \text{AddThirdToWin}$$

# Behavioral Programming

## Tic-Tac-Toe

---

5. *BlockOpponentFromWinning*: Block the other player if they're about to win

$$v5 \triangleq v4$$

$$Init5 \triangleq \text{TRUE}$$

$$Next5 \triangleq CanWin(Opponent) \wedge \neg Priority1 \Rightarrow \\ \exists line \in Line : Count(Opponent, line) = N - 1 \wedge MarkLast(line)$$

$$Priority2 \triangleq Priority1 \vee CanWin(Opponent)$$

$$BlockOpponentFromWinning \triangleq Init5 \wedge \square[Next5]_{v5}$$

$$TicTacToe5 \triangleq TicTacToe4 \wedge BlockOpponentFromWinning$$

---



# Behavioral Programming

## Tic-Tac-Toe

---

6. *MarkCenterIfAvailable*: Prefer center square

$$CenterSquare \triangleq \langle (N + 1) \div 2, (N + 1) \div 2 \rangle$$

$$CenterFree \triangleq board[CenterSquare] = Empty$$

$$v6 \triangleq v5$$

$$Init6 \triangleq TRUE$$

$$Next6 \triangleq (CenterFree \wedge \neg Priority2) \Rightarrow board'[CenterSquare] = current$$

$$Priority3 \triangleq Priority2 \vee CenterFree$$

$$MarkCenterIfAvailable \triangleq Init6 \wedge \square[Next6]_{v6}$$

$$TicTacToe6 \triangleq TicTacToe4 \wedge MarkCenterIfAvailable$$

Properties we can state at this point:

$$FirstMarksSquare \triangleq turn = 1 \Rightarrow board[CenterSquare] \neq Empty$$

THEOREM  $TicTacToe6 \Rightarrow \square(FirstMarksSquare)$

---

# Behavioral Programming

## Tic-Tac-Toe

7. *MarkCornerIfAvailable*: Prefer corner square

$$\text{CornerSquares} \triangleq \{1, N\} \times \{1, N\}$$

$$\text{CornerFree} \triangleq \exists \text{corner} \in \text{CornerSquares} : \text{board}[\text{corner}] = \text{Empty}$$

$$v7 \triangleq v6$$

$$\text{Init7} \triangleq \text{TRUE}$$

$$\text{Next7} \triangleq (\text{CornerFree} \wedge \neg \text{Priority3}) \Rightarrow$$

$$\exists \text{corner} \in \text{CornerSquares} : \wedge \text{board}[\text{corner}] = \text{Empty}$$

$$\wedge \text{board}'[\text{corner}] = \text{current}$$

$$\text{Priority4} \triangleq \text{Priority3} \vee \text{CornerFree}$$

$$\text{MarkCornerIfAvailable} \triangleq \text{Init7} \wedge \square[\text{Next7}]_{v7}$$

$$\text{TicTacToe7} \triangleq \text{TicTacToe6} \wedge \text{MarkCornerIfAvailable}$$

Properties we can state at this point:

$$\text{SecondMarksCorner} \triangleq \text{turn} = 2 \Rightarrow \exists \text{corner} \in \text{CornerSquares} : \text{board}[\text{corner}] \neq \text{Empty}$$

$$\text{THEOREM } \text{TicTacToe7} \Rightarrow \square(\text{SecondMarksCorner})$$

The tactics are sufficient to always force a draw

$$\text{AlwaysDraw} \triangleq (\text{win} \notin \text{Player})$$

$$\text{THEOREM } \text{TicTacToe7} \Rightarrow \square \text{AlwaysDraw}$$

# Behavioral Programming

## Tic-Tac-Toe

$$Next12 \triangleq Compose(Next1, v1, Next2, v2)$$

$$Next123 \triangleq Compose(Next12, \langle v1, v2 \rangle, Next3, v3)$$

$$Next1234 \triangleq Compose(Next123, \langle v1, v2, v3 \rangle, Next4, v4)$$

$$Next12345 \triangleq Compose(Next1234, \langle v1, v2, v3, v4 \rangle, Next5, v5)$$

$$Next123456 \triangleq Compose(Next12345, \langle v1, v2, v3, v4, v5 \rangle, Next6, v6)$$

$$Next1234567 \triangleq Compose(Next123456, \langle v1, v2, v3, v4, v5, v6 \rangle, Next7, v7)$$

$$vars \triangleq \langle v1, v2, v3, v4, v5, v6, v7 \rangle$$

$$Init \triangleq Init1 \wedge Init2 \wedge Init3 \wedge Init4 \wedge Init5 \wedge Init6 \wedge Init7$$

$$Next \triangleq Next1234567$$

$$TicTacToe0 \triangleq Init \wedge \square[Next]_{vars} \wedge WF_{vars}(Next)$$

$$Terminates \triangleq win \neq Empty$$

THEOREM  $TicTacToe0 \Rightarrow \diamond Terminates$

### Statistics

State space progress (click column header for graph)

Time	Diameter	States Found	Distinct States
00:00:06	10	378	62
00:00:05	8	188	44
00:00:02	0	1	1

# Behavioral Programming

## Resources

<https://pron.github.io/files/TicTacToe.pdf>

- Behavioral Programming Home Page: <http://www.wisdom.weizmann.ac.il/~bprogram/>
- Rethinking Software Systems: A friendly introduction to Behavioral Programming: <https://youtu.be/PW8VdWA0UcA>
- Bridging Specifications and Code: Behavioral Programming with React: <https://vimeo.com/298554103>



# Example II





# Proving Possibility Properties

Leslie Lamport, Theoretical Computer Science, 1998

## Proving Possibility Properties

Leslie Lamport

*Digital Equipment Corporation, 130 Lytton Avenue, Palo Alto, California 94303,  
USA*

---

### Abstract

A method is described for proving “always possibly” properties of specifications in formalisms with linear-time trace semantics. It is shown to be relatively complete for TLA (Temporal Logic of Actions) specifications.

*Key words:* Branching time, linear time, temporal logic.

---

### 1 Introduction

Does proving possibility properties provide any useful information about a system? Why prove that it is possible for a user to press  $q$  on the keyboard and for a  $q$  subsequently to appear on the screen? We know that the user can



# Proving Possibility Properties

Although possibility properties may tell us nothing about a system, we do not reason about a system; we reason about a mathematical model of a system. A possibility property can provide a sanity check on our model. Proving that it is always possible for a *press*(*q*) action to occur tells us something useful about the model. In general, we want to prove that a model allows the occurrence of actions representing events that the system cannot prevent.

# Proving Possibility Properties

Although possibility properties may tell us nothing about a system, we do not reason about a system; we reason about a mathematical model of a system. A possibility property can provide a sanity check on our model. Proving that it is always possible for a *press*(*q*) action to occur tells us something useful about the model. In general, we want to prove that a model allows the occurrence of actions representing events that the system cannot prevent.

To prove  $\mathbf{P}_{\Pi}(P)$ , we find an action  $M$  and a conjunction  $G$  of fairness properties such that

$$Init \wedge \square[N]_v \wedge \diamond\square[M]_v \wedge G \Rightarrow \square\diamond P \quad (4)$$

# ***Checking Possibility Properties***

$$Spec \triangleq Init \wedge \Box[Next]_{vars} \wedge Fairness$$

PROPOSITION  $Init \wedge \Box[Next]_{vars} \wedge \Diamond\Box[M]_{vars} \wedge G \Rightarrow \Box\Diamond P$

# ***Checking Possibility Properties***

$$Init \wedge \square[Next]_{vars} \wedge \color{yellow}\diamond\square[M]_{vars} \wedge G$$

# Checking Possibility Properties

$$Init \wedge \square[Next]_{vars} \wedge \diamond\square[M]_{vars} \wedge G$$

VARIABLE  $t$

$$\begin{aligned} & \wedge Init \wedge \square[Next]_{vars} \\ & \wedge t = \text{FALSE} \wedge \square[t \Rightarrow (M \wedge \text{UNCHANGED } t)]_{\langle vars, t \rangle} \wedge \text{WF}_{\langle vars, t \rangle}(t' = \text{TRUE}) \wedge G \end{aligned}$$

$$\diamond\square[M]_{vars} \equiv \square[t \Rightarrow (M \wedge \text{UNCHANGED } t)]_{\langle vars, t \rangle} \wedge \text{WF}_{\langle vars, t \rangle}(t' = \text{TRUE})$$



# Checking Possibility Properties

VARIABLE  $t$

$$Trigger \triangleq t = \text{FALSE} \wedge t' = \text{TRUE}$$

$$PossibilitySpec \triangleq$$

$$\wedge Init \wedge t = \text{FALSE}$$

$$\wedge \square[\vee \wedge Next$$

$$\wedge t \Rightarrow M$$

$$\wedge \text{UNCHANGED } t$$

$$\vee Trigger \wedge \text{UNCHANGED } vars]_{\langle vars, t \rangle}$$

$$\wedge G \wedge \text{WF}_{\langle vars, t \rangle}(Trigger)$$

PROPOSITION  $PossibilitySpec \Rightarrow \square \diamond P$

# Example III





# Model-Based Trace-Checking

Yvonne Howard et al., UK Software Testing and Research, 2003

## Model-Based Trace-Checking

Yvonne Howard, Stefan Gruner, Andrew M Gravell, Carla Ferreira, Juan Carlos Augusto  
DSSE, Department of Electronics and Computer Science, University of Southampton  
Southampton, SO17 1BJ  
Email: ymh@ecs.soton.ac.uk

### Abstract


*Trace analysis can be a useful way to discover problems in a program under test. Rather than writing a special purpose trace analysis tool, this paper proposes that traces can usefully be analysed by checking them against a formal model using a standard model-checker or else an animator for executable specifications. These techniques are illustrated using a Travel Agent case study implemented in J2EE. We added trace beans to this code that write trace information to a database. The traces are then extracted and converted into a form suitable for analysis by Spin, a popular model-checker, and Pro-B, a model-checker and animator for the B notation. This illustrates the technique, and also the fact that such a system can have a variety of models, in different notations, that capture different features. These experiments have demonstrated that model-based trace-checking is feasible. Future work is focussed on scaling up the approach to larger systems by increasing the level of automation.*

### 1 Introduction

From the tester's perspective, tracing is perhaps considered a last resort. When a program or system crashes, it may be necessary to analyse a trace recorded in a log file which can be tedious. A trace viewer [Helmbold90] can help with this task, but human intuition is still

# Conformance in TLA

Implication is refinement/implementation



$$G \Rightarrow F$$

# Model-Based Trace-Checking in TLA+

MODULE *System*

VARIABLES  $x, y, z, tickTock$   
 $vars \triangleq \langle x, y, z, tickTock \rangle$

$TypeOK \triangleq \wedge x \in Nat$   
 $\wedge y \in Nat$   
 $\wedge z \in Nat$   
 $\wedge tickTock \in \{ "tick", "tock" \}$

$Init \triangleq \wedge x \in 0..9$   
 $\wedge y \in 0..9$   
 $\wedge z = 0$   
 $\wedge tickTock = "tick"$

$Next \triangleq \vee \wedge tickTock = "tick"$   
 $\wedge tickTock' = "tock"$   
 $\wedge z' = x + y$   
 $\wedge UNCHANGED \langle x, y \rangle$   
 $\vee \wedge tickTock = "tock"$   
 $\wedge tickTock' = "tick"$   
 $\wedge x' \in 0..9$   
 $\wedge y' \in 0..9$   
 $\wedge UNCHANGED z$

$Safety \triangleq Init \wedge \square [Next]_{vars}$  Just the safety part of the spec

$Spec \triangleq Safety \wedge WF_{vars}(Next)$

## Statistics

State space progress (click column header for graph)

Time	Diameter	States Found	Distinct States
00:00:03	3	12,000	2,000
00:00:02	0	100	100



# Model-Based Trace-Checking in TLA<sup>+</sup>

Tuples are:  $\langle x, y, z, tickTock \rangle$

$Trace \triangleq \langle \langle 1, 0, 0, \text{"tick"} \rangle, \langle 1, 0, 1, \text{"tock"} \rangle, \langle 1, 1, 1, \text{"tick"} \rangle, \langle 1, 1, 2, \text{"tock"} \rangle, \langle 0, 3, 2, \text{"tick"} \rangle, \langle 0, 3, 3, \text{"tock"} \rangle, \langle 2, 2, 3, \text{"tick"} \rangle, \langle 2, 2, 4, \text{"tock"} \rangle, \langle 3, 2, 4, \text{"tick"} \rangle, \langle 3, 2, 5, \text{"tock"} \rangle, \langle 2, 4, 5, \text{"tick"} \rangle, \langle 2, 4, 6, \text{"tock"} \rangle, \langle 5, 2, 6, \text{"tick"} \rangle, \langle 5, 2, 7, \text{"tock"} \rangle, \langle 4, 4, 7, \text{"tick"} \rangle, \langle 4, 4, 8, \text{"tock"} \rangle, \langle 2, 7, 8, \text{"tick"} \rangle, \langle 2, 7, 9, \text{"tock"} \rangle, \langle 6, 4, 9, \text{"tick"} \rangle, \langle 6, 4, 10, \text{"tock"} \rangle \rangle$

# Model-Based Trace-Checking in TLA<sup>+</sup>

VARIABLES  $x, y, z, tickTock$   
 $Model \triangleq$  INSTANCE  $System$

VARIABLE  $i$  the trace index

$Read \triangleq Model!vars = Trace[i]$

$Init \triangleq i = 1 \wedge Read$

$Next \triangleq i < Len(Trace) \wedge i' = i + 1 \wedge Read'$

$TraceBehavior \triangleq Init \wedge \square[Next]_{\langle Model!vars, i \rangle}$

# Model-Based Trace-Checking in TLA+

VARIABLES  $x, y, z, tickTock$

$Model \triangleq$  INSTANCE  $System$

$vars \triangleq \langle x, y, z, tickTock \rangle$  If we write  $Model!vars$ , TLC complains.

VARIABLE  $i$  the trace index

“Reading” a record is just  $vars = Trace[i]$ , but unfortunately TLC isn’t happy with that, so:

$Read \triangleq$  LET  $Rec \triangleq Trace[i]$  IN  $x = Rec[1] \wedge y = Rec[2] \wedge z = Rec[3] \wedge tickTock = Rec[4]$

Unfortunately, TLC also isn’t happy with just  $Read'$  – which is equivalent to:

$ReadNext \triangleq$  LET  $Rec \triangleq Trace[i']$  IN  $x' = Rec[1] \wedge y' = Rec[2] \wedge z' = Rec[3] \wedge tickTock' = Rec[4]$

$Init \triangleq i = 1 \wedge Read$

$Next \triangleq i < Len(Trace) \wedge i' = i + 1 \wedge ReadNext$

$TraceBehavior \triangleq Init \wedge \square[Next]_{\langle vars, i \rangle}$


# Model-Based Trace-Checking in TLA+

*TraceBehavior*  $\Rightarrow$  *Model! Safety*

## Statistics

State space progress (click column header for graph)

Time	Diameter	States Found	Distinct States
00:00:02	20	40	20
00:00:02	0	1	1



# Model-Based Trace-Checking in TLA+

$Traces \triangleq [log1 \mapsto$   
     $\langle\langle 1, 0, 0, \text{"tick"} \rangle, \langle 1, 0, 1, \text{"tock"} \rangle, \langle 1, 1, 1, \text{"tick"} \rangle, \langle 1, 1, 2, \text{"tock"} \rangle,$   
     $\langle 0, 3, 2, \text{"tick"} \rangle, \langle 0, 3, 3, \text{"tock"} \rangle, \langle 2, 2, 3, \text{"tick"} \rangle, \langle 2, 2, 4, \text{"tock"} \rangle,$   
     $\langle 3, 2, 4, \text{"tick"} \rangle, \langle 3, 2, 5, \text{"tock"} \rangle, \langle 2, 4, 5, \text{"tick"} \rangle, \langle 2, 4, 6, \text{"tock"} \rangle\rangle,$   
 $log2 \mapsto$   
     $\langle\langle 5, 2, 0, \text{"tick"} \rangle, \langle 5, 2, 7, \text{"tock"} \rangle, \langle 4, 4, 7, \text{"tick"} \rangle, \langle 4, 4, 8, \text{"tock"} \rangle,$   
     $\langle 2, 7, 8, \text{"tick"} \rangle, \langle 2, 7, 9, \text{"tock"} \rangle, \langle 6, 4, 9, \text{"tick"} \rangle, \langle 6, 4, 10, \text{"tock"} \rangle\rangle,$   
 $log3 \mapsto$   
     $\langle\langle 3, 4, 0, \text{"tick"} \rangle, \langle 3, 4, 7, \text{"tock"} \rangle, \langle 0, 9, 7, \text{"tick"} \rangle, \langle 0, 9, 9, \text{"tock"} \rangle,$   
     $\langle 2, 2, 9, \text{"tick"} \rangle, \langle 2, 2, 4, \text{"tock"} \rangle, \langle 2, 6, 4, \text{"tick"} \rangle, \langle 2, 6, 8, \text{"tock"} \rangle\rangle]$



# Model-Based Trace-Checking in TLA<sup>+</sup>

VARIABLES  $x, y, z, tickTock$   
 $Model \triangleq$  INSTANCE  $System$   
 $vars \triangleq \langle x, y, z, tickTock \rangle$

VARIABLE  $log$ , the  $log$  file  
 $i$  the trace index


$Trace \triangleq Traces[log]$   
 $Read \triangleq$  LET  $Rec \triangleq Trace[i]$  IN  $x = Rec[1] \wedge y = Rec[2] \wedge z = Rec[3] \wedge tickTock = Rec[4]$   
 $ReadNext \triangleq$  LET  $Rec \triangleq Trace[i']$  IN  $x' = Rec[1] \wedge y' = Rec[2] \wedge z' = Rec[3] \wedge tickTock' = Rec[4]$

$Init \triangleq log \in \text{DOMAIN } Traces \wedge i = 1 \wedge Read$   
 $Next \triangleq \wedge i < Len(Trace) \wedge i' = i + 1 \wedge ReadNext$   
 $\wedge \text{UNCHANGED } log$  Each trace follows a single log

$TraceBehavior \triangleq Init \wedge \square[Next]_{\langle log, i, vars \rangle}$

# Model-Based Trace-Checking in TLA+

*TraceBehavior*  $\Rightarrow$  *Model! Safety*



# Model-Based Trace-Checking in TLA<sup>+</sup>

$$\textit{Trace} \triangleq \langle 0, 1, 2, 3, 4, 5, 6, 7, 8, 9, 10 \rangle$$

# Model-Based Trace-Checking in TLA<sup>+</sup>

$$\textit{Trace} \triangleq \langle 0, 1, 2, 3, 4, 5, 6, 7, 8, 9, 10 \rangle$$

$$\textit{TraceBehavior} \Rightarrow \exists x, y, \textit{tickTock} : \textit{Model!Safety}$$

# Model-Based Trace-Checking in TLA<sup>+</sup>

VARIABLES  $z, i$

$tvars \triangleq \langle z, i \rangle$

$Read \triangleq \text{LET } Rec \triangleq Trace[i] \text{ IN } z = Rec$

$ReadNext \triangleq \text{LET } Rec \triangleq Trace[i'] \text{ IN } z' = Rec$

$InitTrace \triangleq i = 1 \wedge Read$

$NextTrace \triangleq i < Len(Trace) \wedge i' = i + 1 \wedge ReadNext$



# Model-Based Trace-Checking in TLA+

VARIABLES  $z, i$   
 $tvars \triangleq \langle z, i \rangle$

$Read \triangleq \text{LET } Rec \triangleq Trace[i] \text{ IN } z = Rec$

$ReadNext \triangleq \text{LET } Rec \triangleq Trace[i'] \text{ IN } z' = Rec$

$InitTrace \triangleq i = 1 \wedge Read$

$NextTrace \triangleq i < Len(Trace) \wedge i' = i + 1 \wedge ReadNext$

VARIABLE  $tt$

$Init \triangleq InitTrace \wedge tt = 0$

$Next \triangleq \vee \wedge i < Len(Trace)$   
 $\wedge tt' = 1 - tt$

$\wedge \vee tt = 0 \wedge NextTrace$

$\vee tt = 1 \wedge \text{UNCHANGED } tvars$


$\vee \text{UNCHANGED } \langle tt, tvars \rangle$  So that we don't get a deadlock error in TLC

$TraceBehavior \triangleq Init \wedge \square[Next]_{\langle tt, z, i \rangle}$

$Model \triangleq \text{INSTANCE } System \text{ WITH } tickTock \leftarrow \text{IF } tt = 0 \text{ THEN "tick" ELSE "tock",}$   
 $x \leftarrow \text{IF } tt = 0 \text{ THEN } z \text{ ELSE } z - 1,$   
 $y \leftarrow 1$


# Model-Based Trace-Checking in TLA+

*TraceBehavior*  $\Rightarrow \exists x, y, tickTock : Model!Safety$



# Model-Based Trace-Checking in TLA<sup>+</sup>

$TraceBehavior \Rightarrow \exists x, y, tickTock : Model!Safety$



**NOT A THEOREM**  $Model!Safety \wedge TraceBehavior \equiv FALSE$

**NOT A THEOREM**  $Model!Safety \wedge TraceBehavior \Rightarrow FALSE$

# Model-Based Trace-Checking in TLA<sup>+</sup>

$vars \triangleq \langle x, y, z, tickTock \rangle$

$tvars \triangleq \langle z, i \rangle$

$Model \triangleq \text{INSTANCE } System$

$ComposedSpec \triangleq Model!Safety \wedge TraceBehavior \equiv$

$\wedge Model!Init \wedge InitTrace$

$\wedge \square [Compose(Model!Next, vars, NextTrace, tvars)]_{\langle vars, tvars \rangle}$



# Model-Based Trace-Checking in TLA<sup>+</sup>

$TraceFinished \triangleq i \geq Len(Trace)$

$Check \triangleq ComposedSpec \Rightarrow \Box(\neg TraceFinished)$

# Model-Based Trace-Checking in TL

$$TraceFinished \triangleq i \geq Len(Trace)$$

$$Check \triangleq ComposedSpec \Rightarrow \Box(\neg TraceFinished)$$

**Statistics**

State space progress (click column header for graph)

Time	Diameter	States Found	Distinct States
00:00:02	20	5,704	1,057
00:00:02	0	100	100

**Statistics**

State space progress (click column header for graph)

Time	Diameter	States Found	Distinct States
00:00:03	3	12,000	2,000
00:00:02	0	100	100

Invariant  $\sim$ TraceFinished is violated.

**Error-Trace Exploration**

Error-Trace

Name	Value
<Initial predicate>	State (num = 1)
i	1
tickTock	"tick"
x	0
y	1
z	0
<Action line 377, col 1 t...>	State (num = 2)
i	2
tickTock	"tock"
x	0
y	1
z	1
<Action line 377, col 1 t...>	State (num = 3)
i	2
tickTock	"tick"
x	0
y	2
z	1
<Action line 377, col 1 t...>	State (num = 4)
i	3
tickTock	"tock"
x	0
y	2
z	2
<Action line 377, col 1 t...>	State (num = 5)
i	3
tickTock	"tick"
x	0
y	3
z	2
<Action line 377, col 1 t...>	State (num = 6)
i	4
tickTock	"tock"
x	0
y	3
z	3
<Action line 377, col 1 t...>	State (num = 7)
i	4
tickTock	"tick"

# Model-Based Trace-Checking in TLA<sup>+</sup>

## Resources

<https://pron.github.io/files/Trace.pdf>

<https://github.com/tlaplus/CommunityModules>

<https://github.com/lemmy/BlockingQueue>



# Conjunction Capers

A TLA+ Truffle



Image: Amazon